
Overview
    

Ergo is a parser combinator library written in Elixir. The name 'Ergo' comes from 'ergo' which means 'therefore' which means 'for that reason' which seemed appropriate for a parser.
If you are already familiar with parser combinators, great! If not we have a gentle introduction or see Saša Jurić's talk where he builds up parser combinators from the ground up.
If you want to know why you should (or shouldn't) use Ergo see comparison with other parser combinator libraries.

  
    
  
  How to install Ergo.


Add Ergo to your mix.exs file:
{:ergo, "~> 0.9"}
Then run
mix deps.get


  

    
Intro to Parser Combinators
    

If you prefer to watch to a talk see Saša Jurić build up parser combinators from the ground up
The phrase parser combinator may sound confusing but is in fact deceptively simple if you already understand the notion of functions and function composition.
For example transforming [1, 2, 3, 4, 5] into [2, 4, 6, 8, 10] can be achieved through combining the functions Enum.map and double as follows:
double = fn x -> 2 * x end
Enum.map([1, 2, 3, 4, 5], double)
If you think of a parser as being a certain kind of function that works on a textual input (in the way that double above is a kind of function that works on a number input) then you may see how we can combine parsers to do work.
For example to parse an input such as "12345" we might combine two parsers many and digit. Let's imagine digit is a parser that accepts an input character in the range of '0'..'9' and returns a number, like this:
digit("12345") = 1
digit("2345") = 2
and so on. Now lets imagine that many is a parser that accepts an input and another parser and keeps attempting to match that parser against the input until it doesn't match any further. So, for example:
many("12345", digit) = [1, 2, 3, 4, 5]
It's the "accepts another parser" concept that is key and is what makes many a combinator parser. In the same way as map calls the function double to do work, many calls the parser digit to do work.
The insight is that we can replace digit with any other parser including other combinator parsers. In this way we can combine, and recombine, simpler parsers to form more complex parsers until we have something that can parse our input.
In Ergo we distinguish between terminal parsers that only work on the input and combinator parsers that are parameterised with one or more other parsers (which may themselves be either a terminal or another combinator parser).


  

    
Comparisons with other libraries
    

There are already at least two mature parser combinator libraries for Elixir: NimbleParsec and ExSpirit so why write another one?
First I am learning about writing parsers with parser combinators and implementing my own seemed like a good way to learn more. I was especially motivated by Saša Jurić's talk where he builds up parser combinators from the ground up. Ergo owes a lot to his style.
Second the code of both ExSpirit & NimbleParsec is, given my relatively limited experience of Elixir, a little hard to digest. I wanted to handle errors better than I'd seen in any of their examples (which seemed to ignore this issue) and finding the code hard to understand made it difficult to do that.
So there are some notable differences between Ergo and NP/ExS that perhaps justify its existence:
First, and maybe difficult to spot, is that Ergo is implemented almost exclusively using functions and not macros while ExS/NP both make significant use of macros. My understanding is not sophisticated enough to understand quite what the macros are buying the user of those libraries but I was able to do without them except in one case. I introduced the
lazy() parser combinator to handle parser recursion (I came across this when parsing values which may also be lists
of values).
Second Ergo is built with error handling as one of its priorities. As a newbie I felt the need, so internally, Ergo parsers are not bare functions but Parser structs that assist with debugging. This takes the form of cycle detection
and telemetry. See debugging for more information.
Third, ExS/NP make heavy use of the Elixir |> operator to combine parsers together, for example:
act =
    string("Act")
    |> replace(:act)
    |> ignore(whitespace)
    |> concat(id)
    |> ignore(whitespace)
    |> ignore(char(?{))
    |> ignore(whitespace)
    |> optional(attributes)
    |> ignore(whitespace)
    |> concat(scene)
    |> repeat(ignore(whitespace) |> concat(scene))
    |> ignore(whitespace)
    |> ignore(char(?}))
    |> wrap
while in Ergo style you'd write:
act =
    sequence([
        literal("Act", ast: fn _ast -> :act end),
        ignore(whitespace),
        id,
        ignore(whitespace),
        ignore(char(?{)),
        ignore(whitespace),
        optional(attribtues),
        ignore(whitespace),
        many(scene),
        ignore(whitespace),
        ignore(char(?}))
    ])
Perhaps unsurprisingly I prefer the Ergo style.
If you need something robust and high-performance I suspect you should be using NimbleParsec rather than Ergo. NP is written by Jose Valim himself and my understanding is that it makes good use of Elixir binary handling for performance.
If, on the other hand, you prefer the Ergo style and would benefit from the debugging support, you could try Ergo and I'd love to hear from you if you do.


  

    
Creating a Basic Parser
    

This tutorial will explore how Ergo works through creating a parser to match numbers, be they integers like -5 or 42, or decimals like 25.6. The parser will match them and convert them into the appropriate Elixir numeric value. We will build it up in stages, starting with bare digits.

  
    
  
  Parsing a digit


To begin with we need to be able to parse digits. One option for that is to use the basic char parser to match digit characters. Using IEx here is what you would do:
alias Ergo
alias Ergo.Context
import Ergo.Terminals

digit = char(?0..?9)
Ergo.parse(digit, "42")
%Context{status: :ok, ast: 52}
Where the integer 52 is the character code of the digit '4' (Try typing ?4 into the IEx console to see for yourself).
Now we can parse a single digit, how about multiple digits?

  
    
  
  Parsing many digits


 To parse multiple digits we use the many parser in conjunction with the digit parser as follows:
import Ergo.Combinators

digits = many(digit())
Ergo.parse(digits, "42")
%Context{status: :ok, ast: [52, 50]}
In fact you might see '42' in your version of the AST because IEx will try to render the list [52, 50] as a charlist. This is a hangover from Erlang. If you would prefer to see the list add the following to your ~/.iex.exs file:
IEx.configure(inspect: [charlists: :as_lists])
The many combinator parser repeatedly invokes the digit parser to match as many digit characters as possible, generating an AST that is a list of those digits. To transform them into a numeric value we need to apply another function to the AST.
In our case the AST list contains character values of the digits matched, e.g. [52, 50] for the digits ['4', '2'] respectively. Since the digit character '0' has character value of 48 we can turn the characters into digit values by subtracting 48.
 The heavy lifting of transforming digits is done by c_transform below which is pipeline to transform ['4', '2'] -> [{4, 10}, {2, 1}] -> [40, 2] -> 42:
c_transform = fn ast ->
  bases = Stream.unfold(1, fn n -> {n, n * 10} end)
  digits = Enum.map(ast, fn digit -> digit - 48 end)
  Enum.zip(Enum.reverse(digits), bases)
  |> Enum.map(&Tuple.product/1)
  |> Enum.sum
  end

digits = many(digit) |> transform(c_transform)

Ergo.parse(digits, "42")
%Context{status: :ok, ast: 42}
In this case we are applying the transform parser to the many parser. Transform only operates on the AST of the parser it is given by applying a function to it. In this case we could also have used:
digits = many(digit(), ast: c_transform)
As many of the combinator parsers support an optional ctx: or ast: argument as a shortcut.
At this point we can parse positive integers of any length:
Ergo.parse(digits, "918212812783918723")
%Context{status: :ok, ast: 918212812783918723}

  
    
  
  Parsing negative numbers


What about negative values? We need to look for a leading '-' character however, unlike the digits, the minus is optional. Parsing the minus is simple enough:
minus = char(?-)

Ergo.parse(minus, "-")
%Context{status: :ok, ast: 45}
45 is the char value of the char '-'. We can now use the optional combinator to allow a minus to be matched, or not:
minus = optional(char(?-))

Ergo.parse(minus, "-42")
%Context{status: :ok, ast: 45}

Ergo.parse(minus, "42")
%Context{status: :ok, ast: nil}
In the second case the status is :ok meaning the optional parser succeeded, however the ast is nil meaning nothing was matched. Let's make this a bit more useful:
minus = optional(char(?-)) |> transform(fn ast ->
  case ast do
    nil -> 1
    45 -> -1
  end
end)

Ergo.parse(minus, "-42")
%Context{status: :ok, ast: -1}

Ergo.parse(minus, "42")
%Context{status: :ok, ast: 1}

Now when minus matches a '-' it will transform it to the value -1. When it doesn't match anything it will transform it to the value 1. Now let's combine it with the other parser.
integer = sequence([
  minus,
  digits
])
The sequence parser tries to match a list of parser in turn and, if they all match, generates an AST composed of a list of each of their results. Let's see how it works:
Ergo.parse(integer, "1234")
%Context{status: :ok, ast: [1, 1234]}

Ergo.parse(integer, "-5678")
%Context{status: :ok, ast: [-1, 5678]}
So we can see that it's easy to get the right result by simply taking the product of the two values in the AST:
integer = sequence([
  minus,
  digits,
  ],
  ast: &Enum.product/1
)

Ergo.parse(integer, "1234")
%Context{status: :ok, ast: 1234}

Ergo.parse(integer, "-5678")
%Context{status: :ok, ast: -5678}
So far so good. We can now parse positive and negative integers.

  
    
  
  Parsing decimals


If we want to parse decimal numbers as well we need to handle the (optional) mantissa, the digits to the right of the decimal point.
We can see that the mantissa is structurally the same as the integer part, a set of digits, but will need to be processed a little differently.
The m_transform function below should look familiar. It works the same way as the c_transform only instead of multiplying by increasing powers of 10, we're dividing by increasing powers of 10.
m_transform = fn ast ->
  ast
  |> Enum.map(fn digit -> digit - 48 end)
  |> Enum.zip(Stream.unfold(0.1, fn n -> {n, n / 10} end))
  |> Enum.map(&Tuple.product/1)
  |> Enum.sum
end

mantissa = many(digit, ast: m_transform)

Ergo.parse(mantissa, "5")
%Context{status: :ok, ast: 0.5}

Ergo.parse(mantissa, "42")
%Context{status: :ok, ast: 0.42000000000000004}
There may be a precision issue with this code but you can see the principle it is operating by.
Now to join the two components together, assuming there is a decimal point (suggesting we'll need optional again). Also we'll again make use of the ast: feature of the sequence combinator to process AST's to give us the right value.
number = sequence([
  integer,
  optional(
    sequence([
      ignore(char(?.)),
      mantissa
    ], ast: &List.first/1)
  )
], ast: &Enum.sum/1)

Ergo.parse(number, "42")
%Context{status: :ok, ast: 42}

Ergo.parse(number, "0.45")
%Context{status: :ok, ast: 0.45}

Ergo.parse(number, "-42")
%Context{status: :ok, ast: -42}
All looking good, just one more example:
Ergo.parse(number, "-4.2")
%Context{status: :ok, ast: -3.8}
Oops! There is a problem with our implementation in that we add together the integer and decimal parts. This works for positive numbers but in the latter case -4 + 0.2 = -3.8 not -4.2. When the integer part is negative we need to subtract the decimal part. We can no longer just use Enum.sum to process the result of the top-level sequence. Instead:
combine = fn
  [integer, decimal | []] ->
    if integer >= 0 do
      integer + decimal
    else
      integer - decimal
    end
  ast ->
    Enum.sum(ast)
end

number = sequence([
  integer,
  optional(
    sequence([
      ignore(char(?.)),
      mantissa
    ], ast: &List.first/1)
  )
], ast: combine)

Ergo.parse(number, "-4.2")
%Context{status: :ok, ast: -4.2}

  
    
  
  Conclusion


Through a series of steps we have built a parser that can handle any kind of integer or decimal number we throw at it. We've seen the using of terminal parsers like char as well as combinator parsers like optional, ignore, many, and sequence and meta parsers like transform (and that often transform can be specified as a ast: argument to a combinator parser).
Hopefully this guide will be helpful in thinking about how to build your own parsers.


  

    
Debugging
    

As parsers become more complex it can be difficult to work out why they fail to operate as expected for a given input,
even a stream of events can be difficult to work with. Ergo parses emit telemetry as they run that is then threaded into
an outline. In this was the debugging information is nested as per the parsers themselves and uninteresting information
is easily hidden to aid focusing on the output that matters.
The telemetry is implemented using the Telemetry package which is quite commonly
used for this purpose by other library's (e.g. Ecto). By default telemetry events are ignored. In order to save them
start the Telemetry service and get the events once parsing is complete.
Ergo.Telemetry.start()
%{status: :ok, id: id} = Ergo.parse(…)
Ergo.Telemetry.get_events(id)
The events can be inspected as a list. However it may be more useful to convert into an outline, by default an OPML
document that can be loaded into most outliners.
File.write("debugging.opml", Ergo.Outline.OPML.generate_opml("My ID", events))
Cycles
A challenge when building parsers is accidentally creating a cycle where the parser will never finish but loop over the same input forever. For example:
many(choice([many(ws()), char(?})]))
Given an input like "}}}" will never finish. The inner many clause will always succeed with 0 whitespace characters and never actually process the char parser at all. You wouldn't deliberately set out to write such a parser but it can happen.
For this reason Ergo implements cycle detection. If the same parser is run a second time on the same input we know we have hit a cycle and an Ergo.Context.CycleError will be raised.


  

    
Recursion
    


  
    
  
  Left Recursion


Parser combinators, being a form of recursive descent parser, are unable to
handle left recursion.
In general this looks like:
A → A𝛼 | β
Where non-terminal A ends up being substituted for itself before any token
can be matched, leading to A → A𝛼 | β again, and so on and so on in an
infinite recursion.
Such parsers can usually be rewritten to a form that is not left-recursive
however that is beyond the scope of this guide.

  
    
  
  Eager Recursion


There is a further problem that arises from the way parser combinators are
defined as functions returning in a language that is not, by default, lazy.
Here is an example parser that is designed to parse values like 42, true,
"What is six times seven?" and also lists of values, including other lists.
Note that we elide handling white space for brevity.
def value() do
  choice([
    number_val(),
    string_val(),
    boolean_val(),
    list_val()
  ])
end

def list_val() do
  sequence([
    char(?[)
    value(),
    many(
      sequence([
        comma(),
        value()
      ])
    ),
    char(?])
  ])
end
A value can be a number, string, boolean, or list. But list is a sequence of
values. The problem arises from a call to value() leading to a call to list()
which, in turn, leads to a call to value(), leading to a call to list() and
so on in an infinite recursion.
Note that this is not grammatical recursion as in the left recursion example
above and cannot be solved by rewriting. The issue is that Elixir evaluates
function calls eagerly, i.e. when they are encountered. We need to introduce
a "call gap" to break the recursion.
We do this with the lazy parser combinator. Lazy wraps it's parser in a
function so that it is not called immediately, breaking the recursion. E.g.:
def value() do
  choice([
    number_val(),
    string_val(),
    boolean_val(),
    lazy(list_val())
  ])
end


  

    
General Guidance
    


  
    
  
  Top-down or Bottom-up?


In my experience when you already have an idea of what it is you want to parse there is a temptation to start at the top and work down. Whenever I do this I seem to run into trouble with something that doesn't fit together right or parsing issues that I then find hard to track down.
My experience is that it's easiest to work bottom up. Parse small actual pieces of the language you want to parse and join these together, building towards the whole format. In general I tend to find I hit parsing errors quicker and can resolve them more easily this way.
While you might have a different experience my advice, if you don't have an opinion yet, is to build your parser bottom up.

  Getting Help


If you decide to use Ergo and you want help please come find me in the Elixir Discord (where I am sandbags). Regardless, I would love to hear from you about problems and or suggestions for improvement.
What is Ergo?
Ergo is an Elixir language parser combinator library. The name 'ergo' means 'therefore' means 'for that reason' which seemed appropriate for a parser.
Also note that this is the second attempt I've made at building such a library. The first, Epic, was not completed and, on reflection, not good maintanable code. I did, however, learn a great deal in building it and that learning has put Ergo on a much firmer footing.
Using Ergo

  
    
  
  Installing


Add Ergo to your mix.exs file:
{:ergo, "~> 0.9"}
Then run
mix deps.get

  
    
  
  A Basic Parser


See the guide to parsing.

  
    
  
  What is a parser?


As with most parser combinator libraries, Ergo parsers are anonymous functions parameterized by inputs. However Ergo parser functions are wrapped in an Ergo.Parser struct that holds the parser function and can contain additional descriptive metadata (by default a description of the parser behaviour).

  
    
  
  The Context


Ergo parsers operate on a Context struct that gets passed from parser to parser. We can create the context directly:
alias Ergo.Context
ctx = Context.new("string to be parsed")
Although you will rarely need to do so as there is a helper function Ergo.parse that will do that for you:
Ergo.parse(parser, "string to be parsed")
Ergo can send debugging information to the Elixir Logger which can be helpful in figuring out why a parser is not working as expected. E.g.
Ergo.parse(parser, "string to be parsed", debug: true)
The status field of the context will always be either :ok or a tuple of {:error, :error_atom}.
The parser can build up a datastructure to return by modifying the ast value of the context. The sequence parser, for example, creates a list of all matching parsers, e.g.
Ergo.parse(sequence([uint(), ignore(char(?,)), uint()]), "1,3")
=> %Context{status: :ok, ast: [1, 3]}

  
    
  
  The Context in detail


At the heart of Ergo is the Context record. All Ergo parser functions take and return a Context. A Context has the following fields:
status
message
input
index
line
col
char
ast

  
    
  
  status


Every parser sets its status to either :ok or a tuple whose first element is the atom :error and whose second element is a list of {code, message} tuples
describing the error in more detail for example {:error, [{:unexpected_char, "Expected 'A' but found '1'}]}.

  
    
  
  input


A binary containing the input remaining to be matched.

  
    
  
  index


The index of the next character to be matched in the input.

  
    
  
  line


The current line of the input the parser has reached.

  
    
  
  col


The current column of the input line that the parser has reached.

  
    
  
  ast


The data structure that the parser constructs from its input.
Parsers
Ergo comes with a set of basic parsers with which you can assemble your own, more complex, parsers. These are in the form of terminal parsers, combinator parsers, and numeric parsers.

  
    
  
  Terminal Parsers


The terminal parsers are not parameterized by another parser and, in general, are low-level parsers for matching specified characters or literal sequences.

  
    
  
  eoi()


The eoi parser returns :ok if the input is empty otherwise returns {:error, unexpected_eoi}.

  
    
  
  char(?c)


The char parser when given a single character code matches that character. If successful it returns a context with status: :ok with the char and ast parameters set to the character c. Otherwise it returns status: {:error, :unexpected_char}
Example: char(?a) matches the character 'a'

  
    
  
  char(?l..?h)


The char parser when given a range of characters. If successful it returns a context with status: :ok with the char and ast parameters set to the matched character. Otherwise it returns status: {:error, :unexpected_char}
Example: char(?A..?Z) matches any uppercase letter

  
    
  
  char(-?c)


The char parser when given a negative character matches any character except the specified character and return :ok with the char and ast parameters set to the matched character. Otherwise it returns status: {:error, :unexpected_char}
Example: char(-?,) matches any character except a comma

  
    
  
  char([...])


The char parser when given a list matches a character according to the specifications in the list which can be either ?c, -?c, or [?l..?h]. If successful it returns a context with status: :ok with the char and ast parameters set to the matched character. Otherwise it returns status: {:error, :unexpected_char}
Example: char([a..z, 5]) matches any lower case letter or the number 5

  
    
  
  digit


The digit parser matches a character in the range [0..9] from the input. If successful it returns a context with status: :ok with the char and ast parameters set to the character code of the digit that has been matched. Otherwise it returns status: {:error, :unexpected_char}

  
    
  
  alpha


The alpha parser matches a character in the range [a..z, A..Z] from the input. If successful it returns a context with status: :ok with the char and ast parameters set to the character code of the alpha character that has been matched. Otherwise it returns status: {:error, :unexpected_char}

  
    
  
  wc


The wc parser matches a word character from the input (it is equivalent to the \w regular expression). If successful it returns a context with status: :ok with the char and ast parameters set to the character code of the alpha character that has been matched. Otherwise it returns status: {:error, :unexpected_char}.

  
    
  
  ws


The ws parser matches a whitespace character from the input (it is equivalent to the \s regular expression). If successful it returns a context with status: :ok with the char and ast parameters set to the character code of the alpha character that has been matched. Otherwise it returns status: {:error, :unexpected_char}.

  
    
  
  literal(s)


The literal parser is given a binary string and attempts to match it against the input. If successful it returns a context with status: :ok with the char and ast parameters set to the string being matched. Otherwise it returns {:error, :unexpected_char} for the first character that doesn't match the input.
Example: literal("Ergo") matches the characters 'E', 'r', 'g', and 'o' from the input successively.

  
    
  
  Numeric parsers


These are parsers that build on the terminal parsers to parse diffrent types of numeric values.
The utility parsers are technically terminal parsers, since they are not parameterised by a parser, but they are implemented in terms of some of the combinator parsers themselves and hence belong in a separate category.

  
    
  
  digits


The digits parser matches a series of digits from the input. If successful it returns with status: :ok and the ast set to a list of the digit values.

  
    
  
  uint


The uint parser matches a series of digits from the input. If successful it returns with status: :ok and ast parameter set to the integer value of the number matched.

  
    
  
  decimal


The decimal parser matches a series of digits, separated by a single '.' from the input. If successful it returns wtih status: :ok and ast set to the floating point value of the number matched.

  
    
  
  number


The number parser builds upon the uint and decimal parsers to parse any kind of numeric value returning status: :ok and ast set to the integer or floating value of the number parsed.

  
    
  
  Combinator Parsers


The combinator parsers are structural parsers that are parameterised with other parsers to match more complex structures.

  
    
  
  sequence


The sequence/2 parser is used to match a given set of parsers in sequence.
Example:
p = sequence([literal("Hello"), char(?\s), literal("World")])
p.(Context.new("Hello World")) => %{status: :ok, ast: ["Hello", ' ', "World"]}

The parser p will match first the literal "Hello" then a single space and then the literal "World". If any of these parsers fail, the sequence parser will fail. This example would more simply be written literal("Hello World") but more complex examples will use other combinator parsers in the sequence.
Optional arguments:
ast: fn ast -> ast
ctx: fn ctx -> ctx

Provide a function that transforms the sequence ast into another form. For example to transform the elements of the list ast into a record type.

debug: true|false

When the sequence parser runs it logs debugging information.

label: <<binary>>

Provide a label that can be logged by setting debug: true

  
    
  
  choice


The choice/2 parser is used to match one from a sequence of parsers. It attempts to match the input against each parser in turn. The first parser that matches will cause choice to succeed with the ast set to the ast of the matching parser. If all of the parsers fail to match the input then choice will fail.
Signature:
choice([parser1 | [parser2 | ...]], args \\ [])
Example:
    p = choice([literal("Hello"), literal("World")])
    p.(Context.new("Hello World")) -> %{status: :ok, ast: "Hello", input: " World"}
    p.(Context.new("World Hello")) -> %{status: :ok, ast: "World", input: " Hello"}

    boolean = choice([literal("true"), literal("false")], ast: fn ast -> ast == "true" end)
    boolean.(Context.new("true")) -> %{status: ok, ast: true, input: ""}
    boolean.(Context.new("false)) -> %{status: ok, ast: false, input: ""}
    boolean.(Context.new("Hello)) -> %{status: {:error, ...}, input: "Hello"}

  
    
  
  many


The many/2 parser is used to match another parser repeatedly on the input. In its default form that can be zero times (i.e. no matches) or infinite matches. However using the :min and :max optional arguments allow limits to be specified.
Signature:
many(parser, args \\ [])
Example:
    p = many(char(?\s), min: 1)
    p.(Context.new("    ")) -> %{status: :ok, input: ""}

    p = many(char(?\s), min: 2)
    p.(Context.new(" Hello")) -> %{status: {:error, ...}, input: " Hello"}

    p = many(char(?a..?z), max: 2)
    p.(Context.new("Hello World)) -> %{status: :ok, ast: ['H', 'e'], input: "llo World"}

    p = many(char(?a..?z), ast: fn ast -> Enum.count(ast) end)
    p.(Context.new("Hello")) -> %{status: :ok, ast: 5, input: ""}

  
    
  
  optional


The optional/1 parser is used to match another parser on the input and succeeds if it doesn't match or matches once.

  
    
  
  ignore


The ignore/1 parser is designed for use with parsers such as sequence and many. It takes a parser and attempts to match it on the input however, if it succeeds, it returns a nil ast value that parsers can use to ignore it.

  
    
  
  transform


The transform/2 parser is used to change the ast returned by another parser if the other parser is successful. It is useful for modifying the output of parsers that do not directly support a map argument.
Signature:
squared = transform(uint(), fn ast -> ast * ast end)
squared.(Context.new("10")) -> %{status: :ok, ast: 100}

  
    
  
  lookahead


The lookahead/1 parser attempts to match the parser it is given on the input. If it succeeds it returns with status: :ok but ast: nil and with the input unchanged. Otherwise it returns with status: {:error, :lookahead_fail}.
p = lookahead(literal("Hello"))
p.(Context.new("Hello World")) -> %{status: :ok, input: "Hello World"}

  
    
  
  not_lookahed


The not_lookahead/1 parser is the inverse of the looakahead parser in that it attempts to match its parser on the input and where it can do so it returns with status: {:error, :lookahead_fail}.
p = not_lookhead(literal("Hello"))
p.(Context.new("Hello World")) -> %{status: {:error, :lookahead_fail}}

Ergo.Combinators is the key set of parsers used for combining together other parsers.
Parsers
	choice
	sequence
	many
	optional
	ignore
	transform
	lookeahead
	not_lookahead

Ergo.Context defines the Context struct that is used to maintain parser state as the various
parsers work, and various functions for creating & manipulating contexts.
Fields
	status

When a parser returns it either sets status to :ok to indicate that it was successful or to a tuple
{:error, reasons} where reasons is a list of tuples representing a sequence of errors. Each tuple is of
the form {code, message} where code is an atom indiciting the specific type of error and the message
contains additional human-focused information that may be helpful in diagnosing the problem.
	input

The binary input being parsed.
	consumed

The binary input that has already been consumed by parsing.
	index

Represents the position in the input which has been read so far. Initially 0 it increments for each character processed.
	line

Represents the current line of the input. Initially 1 it increments whenever a  is read from the input.
	col

Represents the current column of the input. Initially 1 it is incremented every time a character is read from the input and automatically resets whenever a  is read.
	entry_points

A list of Parser entry points
	data

Map containing user-data that the parsers can use to pass information between them.
	ast

Represents the current data structure being built from the input.
	parser

The current parser.
	tracks

Parsers will add themselves to the Context tracks in the form of {ref, index}.
If the same parser attempts to add itself a second time at the same index an
error is thrown because a cycle has been detected.
	depth

As nested parsers are called the depth field will be updated to reflect the
number of levels of nesting for the current parser.
	captures

A map of data that can be captured from ASTs using the capture parser.

      


      
        Summary


  
    Functions
  


    
      
        
  
    
    Ergo.Meta - Ergo v1.0.3
    
    

    


  
  

    
Ergo.Meta 
    



      
The Meta parsers are not really parsers at all but operate within the parsing framework.
All Meta parsers are combinators that accept a parser and either report on or modify
its operation.

      


      
        Summary


  
    Functions
  


    
      
        
  
    
    Ergo.Numeric - Ergo v1.0.3
    
    

    


  
  

    
Ergo.Numeric 
    



      
The Parsers module exists to house utility parsers that while they are terminals in the sense that they are not parameterised, they internally make use of parsers from the Combinators module.
Parsers
	uint
	decimal
	digits

Ergo.Parser contains the Parser record type. Ergo parsers are anonymous functions but we embed
them in a Parser record that can hold arbitrary metadata. The primary use for the metadata is
the storage of debugging information.

Ergo.Terminals contains the terminal parsers, which are those parsers not
parameterized with other parsers and therefore work more at the level of text
than structure.

